


clear all
set more off
pause off
set matsize 11000

capture confirm file "Appendix_Results/"
if _rc mkdir "Appendix_Results/"


capture confirm file "Appendix_Results/gphs/"
if _rc mkdir "Appendix_Results/gphs/"




local region  Euro 
local response R_real



use "Data/Big_Pandemics.dta", clear
gen pandemic=1 
sort year
gen midyear = year + (endyear - year)/2
replace midyear = round(midyear,1)
drop year
ren midyear year
merge 1:1 year using "Data/RealRateSince1310.dta"
sort year
replace pandemic=0 if pandemic==.
gen pandemic100k=cond(dead>=100000,pandemic,0)
cap drop Horizon
gen Horizon = _n-1 if _n<=41
tsset year


cap drop b0* se0* 
cap drop h0* l0* h1* l1*
tsset year
foreach x of local region {
	foreach y of local response {
	cap drop `x'`y'random
	cap drop `x'`y'c
	gen b0_`x'`y'=0 if Horizon==0
	gen se0_`x'`y'=0 if Horizon==0	
	gen h0_`x'`y'=0 if Horizon==0
	gen l0_`x'`y'=0 if Horizon==0
	gen h1_`x'`y'=0 if Horizon==0
	gen l1_`x'`y'=0 if Horizon==0

	
	cap drop raw
gen raw = .
replace raw = `x'`y'
/********** State space specification starts here *************/

constraint 1	[rstar]L.rstar		= 1		/* Random Walk */	
constraint 2	[raw]rstar			= 1		/* Definitional */


cap drop zero								/* generate zero line for figures */
gen zero = 0

tsset year
cap drop r_star

/************	 State space equations	 ******************/

sspace		(rstar L.rstar, state noconstant)		/// /* r-star */
			(raw rstar, noconstant ),				/// /* raw */
			covstate(identity) constraints(1/2) difficult /*method(kdiffuse)*/ iterate(25)
			
			qui{
			/* Generate smoothed estimates of r-star */
			predict r_star if e(sample), states smethod(smooth) equation(rstar)
			
			
	
			cap drop `x'`y'random
			gen	`x'`y'random = r_star

			}


								}
								
						}
						
						

			
			
			
local region   Euro
foreach y of local response {
	foreach x of local region {
	eststo clear
		forv h=0/40 {
		
						cap drop  d`y'`h'
						gen d`y'`h' = f`h'.`x'`y'random- l.`x'`y'random
						reg d`y'`h' pandemic100k l(0/10).pandemic100k  l(2/11).`x'`y'random 
												
						eststo
						replace b0_`x'`y'  = _b[pandemic100k]                      if Horizon==`h'
						replace se0_`x'`y' = _se[pandemic100k]                     if Horizon==`h'
						replace h0_`x'`y'  = _b[pandemic100k]+2*_se[pandemic100k]  if Horizon==`h'
						replace l0_`x'`y'  = _b[pandemic100k]-2*_se[pandemic100k]  if Horizon==`h'
						replace h1_`x'`y'  = _b[pandemic100k]+1*_se[pandemic100k]  if Horizon==`h'
						replace l1_`x'`y'  = _b[pandemic100k]-1*_se[pandemic100k]  if Horizon==`h'
						}
		
		esttab , keep(pandemic*)
		cap drop Zero
		gen Zero=0
		twoway ///
		(  rarea h0_`x'`y' l0_`x'`y' Horizon if Horizon~=. , fc(blue)  fi(10) lcolor(white)  lpattern(none)  )  ///
		(  rarea h1_`x'`y' l1_`x'`y' Horizon if Horizon~=. , fc(blue)  fi(20) lcolor(white) lw(none)  lpattern(none)  )  ///
		(  line  Zero    Horizon if Horizon~=. , lc(gray)  lw(med) lp(dash)                     )  ///		
		(  line  b0_`x'`y'     Horizon if Horizon~=. , lc(blue)  lw(thick) lp(solid)  ///
				xlab(0(10)40, labsize(large)) ylab(, angle(0) labsize(large)  nogrid) xtitle("Years since pandemic's middle year  ",  size(large)) ///
				ytitle("Percent",  size(large))  ///
				legend( off ) )  ///
		, title()    ///
		xsize(5) ysize(4) scale(1.1) scheme(s1color) graphregion(color(white)) plotregion(color(white)) ///
		name(`x'`y'_midyear,replace)  
		
graph save   Appendix_Results/gphs/FigureA3b.gph, replace
graph export Appendix_Results/FigureA3b.pdf, replace
		

}
}


